

WORKER BURNOUT: A DYNAMIC MODEL WITH IMPLICATIONS
FOR PREVENTION AND CONTROL

Jack B. Homer
9215 Jellico Ave.
Northridge, CA 91325
USA

ABSTRACT

This paper explores the dynamics of "worker burnout", a process in which a hard-working individual becomes increasingly exhausted, frustrated, and unproductive. The author's own two-year experience with repeated cycles of burnout is qualitatively reproduced by a small system dynamics model which portrays the underlying psychology of "workaholicism". Model tests demonstrate that the limit cycle seen in the base run can be stabilized through techniques which diminish work-related stress or enhance relaxation. These stabilizing techniques also serve to raise overall productivity, since they support a higher level of energy and more working hours on average. One important policy lever is the maximum workweek or "work limit"; an "optimal work limit" at which overall productivity is at its peak is shown to exist within a region of stability where burnout is avoided. The paper concludes with a strategy for preventing burnout which emphasizes the individual's responsibility for understanding the self-inflicted nature of this problem and pursuing an effective course of stability.

BACKGROUND

"In this world there are two tragedies. One is not getting what one wants, the other is getting it." [1]

"The survival of mankind as a species may well depend on the successful management of stress." [2]

The negative consequences of unrelieved stress have become increasingly evident over the last few decades. In a fast-moving, achievement-oriented society, individuals may easily

454

become worn down and unable to function effectively if they are not careful. Excessive stress can lead not only to chronic fatigue, but also to a wide variety of psychological, medical, and behavioral problems ranging from irritation, depression, and loss of appetite to violence, alcoholism, mental illness, and heart disease. [3] These problems affect both the individual and the society at large and have had a clear depressive effect on economic productivity in the United States. It has been estimated that output could be boosted by at least ten percent if the work loss and impaired job performance attributable to mismanaged stress were eliminated. [4] As most executives surely now realize, the connection between distress and the bottom line is indeed real.

Since work plays a central role in the lives of most people, it should not be surprising that a great deal of potentially harmful stress originates in the workplace. In recent years, many researchers in the area of occupational safety and health have turned their attention to the issue of occupational stress. Potential sources of stress at work are many and varied and include: irritants in the physical or social environment, problems of role or responsibility, poor job fit, inadequate rewards or support, and deadline pressure. [5] Some jobs tend to be more stressful than others; studies have shown, for example, that air traffic controllers are much more susceptible to stress-related diseases and behavioral problems than the average person. The same is probably true for lawyers, doctors, social workers, and salespeople. [6] But because tension

may arise in any situation where personal needs are not being satisfied, every job has the potential for being stressful.[7]

When work-related stress becomes severely debilitating, the affected individual may be said to be "burned out". Burnout has been defined broadly as "a process in which individuals become exhausted by making excessive demands on energy strength." [8] It is generally agreed that at the root of the problem is the individual's own overcommitment to frustrating work. The source of a burnout victim's frustration is the inability to attain high expectations set by others or, more frequently, by the individual himself. The entire process may take weeks, months, or years. [9] Some have said that organizations or even entire societies may burn out if they push themselves too hard. [10]

The burnout process begins when the individual attempts to meet unmet expectations by working longer hours. Longer hours mean more exposure to the normal stress of work, and consequently, more of a drain on the individual's finite store of "adaptation energy" and less time available for recovery of that lost energy. [11] This drain of energy may, in turn, render the individual weaker and less capable of reaching his goals. The response to continued inadequacy of performance is to work harder, which depletes energy further. [12] In addition, the worker's growing frustration at work increases the very stressfulness of that work, so that energy is drained still more rapidly. [13] If the individual refuses or is unable to take time out to recover, the vicious cycle of frustration-exhaustion-

455

dysfunction will ultimately produce chronic and severe problems which force him off the job, a burned-out ember of his former self. [14]

The ideal context for burnout combines a "workaholic" personality type with a disagreeable, and therefore highly stressful, job. In general, a non-workaholic will simply be unwilling to work unusually long and stressful hours and certainly will not seek out such a position. Workaholics or high achievers, on the other hand, create conditions of work overload for themselves and feel uneasy when they are not working. [15] They compete most strongly with themselves, essentially expanding their goals whenever necessary to maintain the challenge they inherently need; current accomplishments are never quite enough. [16] The fact that most workaholics can push themselves this way for many years without burning out speaks well for their ability to find work that is so fulfilling as to minimize the wear and tear of long hours. [17] But put a self-driven workaholic in a disagreeable work setting--for example, one lacking in tangible rewards--and the prescription for burnout is complete: Unmet goals become more of a burden than a motivating challenge. [18]

It is worth looking more closely at the workaholic personality. Workaholism is a way of life for perhaps five percent of American adults. Beneath their energetic and intense surface lies an obsessive perfectionism associated with the fear of failure or boredom. [19] Their inability to relax, compromise,

seek assistance, or admit limitations often arises first in childhood, reflecting in part the internalization of family values regarding the importance of determined effort, self-reliance, and self-improvement.[20] But the relatively recent phenomenon of workaholism is not attributable solely to the American work ethic, which, after all, goes back many generations. The difference in America today is that, with the rapid and widespread erosion of our traditions and support systems, few activities outside of the workplace continue to give the individual the sense of meaning and control that they once did. As a result, "...our expectations from work become disproportionate to what, in most cases, it can provide." [21] Since the workaholic invests little of himself outside work, he risks losing all if work itself ceases to be rewarding.

Strategies for the prevention and amelioration of burnout are as numerous as the writers who propose them. But they all seem to fall into one of three categories: (1) Work less, (2) Minimize the stress of work, and (3) Relax more effectively when not working. Working less may involve working fewer hours per day or taking more frequent vacations.[22] Minimizing work stress may involve shifting tasks or reducing expectations, or learning through techniques such as assertiveness training and "stress inoculation" consciously to defuse and redirect potentially stressful situations.[23] Minimizing stress also may require action by the supervisor or employer, such as clarifying goals or roles, providing more rewards and "strokes", allowing for more personal initiative, and generally improving the work

climate.[24] More effective relaxation may involve exercise, hobbies, loafing, close friendship, meditation, biofeedback, body realignment therapy, or any of several techniques specifically designed to elicit the calm state of "passive attention".[25]

Because the sources of stress are manifold and varied, there exists no single panacea for this complex problem. But each of the above techniques can be helpful; in contrast, palliatives which are themselves addictive, such as drinking and gambling, clearly are not effective in managing stress and may only worsen the situation.[26] The prevention of burnout ultimately depends on maintaining a healthy balance of activities, giving adequate time to both "active coping" and rest.[27]

While the literature provides graphic descriptions of the process of burning out and offers many pieces of advice, it does little to address the question of what actually happens following burnout, that is, during and after recovery. Presumably, the individual can regain his lost energy during the time off from work, but what then? If he returns to stressful work following recovery, without adopting measures to manage his energy more effectively, another round of burnout would seem likely.

Although repeated cycles of burnout are apparently undocumented in the literature, they were, in fact, my experience from 1981 to 1983, the two years of researching and writing my doctoral thesis. Perhaps four times during this long and often lonely effort, I experienced periods of moderate burnout, the

symptoms including exhaustion, confusion, anger, headaches, stomach pains, and depression. My productivity plummeted during these periods and I would finally take time off to recover my strength. When I returned to work, I would not only feel refreshed but also have a more relaxed attitude toward work, with lower expectations for my weekly output. I could meet these new goals without working too hard, but soon I would find my expectations rising and work hours increasing, true to the workaholic profile. First, my evenings disappeared, then my weekends, and it was only a matter of time before I would start to feel tired and confused again.

PURPOSE AND APPROACH

The purpose of this paper is to present a dynamic model of worker burnout which is consistent with both the literature and my personal experience, and which can suggest guidelines for attaining greater stability and higher productivity. The model will be tested to determine whether stabilizing policies do in fact raise overall accomplishment; uncertainty on this point arises because the ways of the burnout candidate are precisely those of the high achiever. Can one actually push less and achieve more?

The next section describes the structure of the small (four-level, nineteen-equation) system dynamics model used for the subsequent analysis, first in overview and then in detail. This is followed by an investigation of model behavior: First, a description and causal-loop explanation of the base run; and

second, an exploration of policies to prevent and control burnout cycles, culminating in the idea of an "optimal work limit". The paper concludes with a general strategy for fighting worker burnout, based on insights provided by the model.

MODEL STRUCTURE

Overview

The model of worker burnout to be presented here focuses on the psychological dynamics underlying the problem, namely, the dynamics of workaholism. (See the Appendix for a complete model listing.) The model has four major elements: (1) Accomplishments per week, the outcome measure of interest; (2) Expected accomplishments per week, the outcome goal set by the worker; (3) Hours worked per week, also adjusted by the worker and one determinant of the weekly accomplishment rate; and (4) Energy level, which determines hourly productivity, this being the other factor determining weekly accomplishment. The energy level is depleted by the stress of work and replenished during periods of rest; thus, longer hours at work lead to faster depletion and slower recovery of energy. The frustration that comes from accomplishment falling short of expectations can also accelerate energy depletion. If current accomplishment seems adequate or nearly so, the textbook workaholic will increase his expectations; he allows his goals to drift downward only if they appear far too high given his current output. On the other hand, he will tend to work longer hours if he perceives his accomplishments to be inadequate; and will decrease his hours

only if he has exceeded his goals or if his energy level is so low as to make time off unavoidable.

The model does not attempt to account for the individual's basic approach to and compatibility with his work; thus, the factors which define his workaholic personality are exogenous, as is his normal level of stress. Also beyond the model's boundary are those personal and environmental factors which determine the individual's basic ability to relax. In addition, his hourly output at a given energy level, determined in real life by such factors as his preparation for the task at hand, native intelligence, and assistance from others, is exogenous; the model is not concerned with learning curves or organizational dynamics. This is not to say that changes in the individual or his surroundings are not important; indeed, the role of sensitivity testing is precisely to analyze the importance of various exogenous factors. Rather, the model is intended to show how the workaholic syndrome can by itself lead to burnout, without the aid of other dynamic factors. One need not examine changes in the individual's career goals, for example, in order to explain the month-to-month dynamics of burnout, though such longer-term changes may be central to an explanation of a mid-life "achievement crisis".[28]

Equation Description

A note on parameter values: The values ascribed to constants and table functions in the baseline model described below are entirely based on logic and considered judgment. The

458 numerical data of interest are simply not to be found in the literature, nor have I attempted to measure these parameters myself. The numbers were drawn primarily in an impressionistic fashion from my own experience as a victim of burnout. Note that the burnout cycle reference mode itself was presented above in descriptive, not numerical terms. Thus lacking in numerical data on both structure and behavior, attempts at historical accuracy or prediction are obviously out of the question. Still, if the endogenous structure is potent enough, one can learn much about the generic process through careful model testing. The baseline set of parameters is treated simply as a take-off point for investigating model behavior under a variety of circumstances; the results can thereby be considered applicable to a whole spectrum of individuals and work settings.

- AW.K = (AH.K)(HWW.K) A,1
- AH.K = TABLE(TAH,EL.K,0,1,.2) A,2
- TAH = 0/.2/.4/.6/.8/1 T,2.1
- AW - Accomplishments per Week (A-units/week)
- AH - Accomplishments per Hour (A-units/week)
- HWW - Hours Worked per Week (hours/week)
- TAH - Table for Accomplishments per Hour
- EL - Energy Level (0-1)

The worker's weekly output is the product of his hourly productivity and the number of hours worked per week. Output is measured in units of accomplishment (A-units), the tangible meaning of which depends on the kind of work involved. For simplicity's sake, it has been assumed that the worker can produce at most one accomplishment per hour. Hourly productivity is determined by the worker's energy level: When one's energy

level (and, correspondingly, one's levels of rationality and self-esteem) is low, one is easily distracted from the task at hand and prone to erratic performance and poor decision-making.[29] Indeed, hourly accomplishment might be thought of as an operational definition of the energy level, implying the linear relationship depicted in Equation 2.1. When the worker has no energy at all, he produces nothing, while at full energy he produces at his maximum hourly rate.

HWW.K = HWW.J + (DT/TAHWW)(IHWV.J-HWW.J)	L,3
HWW = HWWI	N,3.1
HWWI = 40	C,3.2
TAHWW = 1	C,3.3
IHWV.K = MIN(LHWW,HWW.K*EELHW.K*EPAHW.K)	A,4
LHWW = 80	C,4.1

- HWW - Hours Worked per Week (hours/week)
- HWWI - Hours Worked per Week, Initial (hrs/wk)
- TAHWW - Time to Adjust HWW (weeks)
- IHWV - Indicated Hours Worked per Week (hrs/wk)
- LHWW - Limit on Hours Worked per Week (hrs/wk)
- EELHW - Effect of Energy Level on Hours Worked
- EPAHW - Effect of Perceived Adequacy on Hours Worked

The number of hours worked per week, initialized at a "standard" value of forty, is adjusted by the worker toward an "indicated" value more in line with his perception of the current situation. It is assumed here that hours are flexible and the adjustment can be made within one week's time. The indicated workweek may be longer than the current workweek if output is perceived as inadequate, and may be shorter if output is perceived as more than adequate or if the energy level is low (see Equations 5 and 6 below); but the workweek is assumed never to exceed some upper limit (or to exceed it so rarely as to be

459

insignificant). This weekly work limit may be determined by the worker or by others, and may be explicit or implicit; in any case, it represents the maximum workweek the individual is willing to put in, on a continuous basis if need be. Many people regularly work sixty to ninety hours per week;[30] the baseline model assumes a work limit of eighty hours per week.

EELHW.K = TABLE(TEELHW,EL.K,0,1,.2)	A,5
TEELHW = 0/.4/.7/.9/1/1	T,5.1
EPAHW.K = TABLE(TEPAHW,PAA.K,0,1.6,.2)	A,6
TEPAHW = 2.3/1.9/1.6/1.35/1.15/1/.9/.8/.75	T,6.1
EELHW - Effect of Energy Level on Hours Worked	
TEELHW - Table for EELHW	
EL - Energy Level	
EPAHW - Effect of Perceived Adequacy on Hours Worked	
TEPAHW - Table for EPAHW	
PAA - Perceived Adequacy of Accomplishment	

Figure 1 shows the "time off" effect which a low energy level may have on hours worked. The function becomes steep only in the region of lower energy, reflecting the workaholic's natural reluctance to break away from work unless the situation is desperate. Workaholics feel guilty and anxious about leaving work and so tend to skip or shortchange vacation time.[31] But should the worker's energy drop toward zero, a vacation will become a matter of necessity rather than choice.

Figure 2 depicts the "work harder" response to performance that falls short of one's goals; studies have shown that the further the goal is, the harder people will work to achieve it.[32] Experience and the logic of symmetry suggest that achievement exceeding one's goals tends to call forth a more




Figure 1. Table for Effect of Energy Level on Hours Worked




Figure 2. Table for Effect of Perceived Adequacy on Hours Worked

relaxed attitude with fewer hours worked. The concave function used here represents a response roughly proportional to the perceived degree of inadequacy or surplus for values of input in the normal range, but considerably less than proportional in the region of very low adequacy; the latter reflects a natural

460 resistance to radical increases in the workweek over a short period of time.

$$PAA.K = PAW.K/XAW.K \quad A,7$$

$$PAW.K = PAW.J + (DT/TPAW)(AW.J-PAW.J) \quad L,8$$

$$PAW = AW \quad N,8.1$$

$$TPAW = 1 \quad C,8.2$$

PAA - Perceived Adequacy of Accomplishment
 PAW - Perceived Accomplishments per Week (A-units/wk)
 XAW - Expected Accomplishments per Week (A-units/wk)
 AW - Accomplishments per Week (A-units/wk)
 TPAW - Time to Perceive Accomplishments per Week (weeks)

The worker's satisfaction with his performance has both psychological and behavioral consequences and reflects a comparison of perceived accomplishment with expected accomplishment; the dimensionless ratio measure used here seems a reasonable approximation to the informal calculus done in real life. The worker assesses his output rate by averaging it over some recent time period; the smoothing time of one week used here reflects the attentive self-observation typical of a high achiever.

$$XAW.K = XAW.J + (DT)(XAW.J*FCXAW.J) \quad L,9$$

$$XAW = XAWI \quad N,9.1$$

$$XAWI = 40 \quad C,9.2$$

$$FCXAW.K = BFCX + FCXPA.K \quad A,10$$

$$BFCX = .1 \quad C,10.1$$

$$FCXPA.K = TABLE(TFCXPA, PAA.K, 0, 1.6, .2) \quad A,11$$

$$TFCXPA = -.7/-0.5/-0.35/-0.2/-0.1/0/.1/.25/.4 \quad T,11.1$$

XAW - Expected Accomplishments per Week (A-units/wk)
 XAWI - XAW, Initial (A-units/wk)
 FCXAW - Fractional Change in XAW (1/wk)
 BFCX - Bias for Fractional Change in Expectations (1/wk)
 FCXPA - Fractional Change in Expectations from Perceived Adequacy (1/wk)
 TFCXPA - Table for FCXPA
 PAA - Perceived Adequacy of Accomplishment

The worker's expectation for weekly output, initialized to equal actual output, is adjusted up or down in response to the perceived adequacy of his performance. The high achiever pushes to expand his goals once they have been met; this is represented in the model by a bias causing the increase of expected output whenever perceived adequacy is neutral (PAA=1). When accomplishment is more than just satisfactory, the worker will feel encouraged to expand his goals even faster than this bias, as shown in Figure 3. Conversely, when output is inadequate, the individual will naturally be tempted to draw back his goals somewhat to avoid undue frustration.[33] In the baseline model, this reaction overcomes the upward bias and causes an actual shrinking of expectations whenever perceived adequacy is less than .8.[34] Figure 3 indicates that this response becomes stronger as the worker's dissatisfaction becomes more acute.




Figure 3. Table for Fractional Change in Expectations from Perceived Adequacy

$EL.K = EL.J + (DT)(ER.JK - ED.JK)$ 12,L
 $EL = ELI$ 12.1,N
 $ELI = 1$ 12.2,C

 $ER.KL = (ERN)(EHWER.K)(EHEFR.K)$ 13,A
 $ERN = .3$ 13.1,C

 $EHWER.K = TABLE(TEHWER, HWW.K, 0, 120, 20)$ 14,A
 $TEHWER = 1.3/1.2/1/.7/.5/.35/.25$ 14.1,T

 $EHEFR.K = TABHL(TEHEFR, EL.K, .8, 1, .05)$ 15,A
 $TEHEFR = 1/.9/.7/.4/0$ 15.1,T

 EL - Energy Level (0-1)
 ELI - Energy Level, Initial (0-1)
 ER - Energy Recovery (1/wk)
 ED - Energy Depletion (1/wk)
 ERN - Energy Recovery Normal (1/wk)
 EHWER - Effect of Hours Worked on Energy Recovery
 TEHWER - Table for EHWER
 HWW - Hours Worked per Week (hours/wk)
 EHEFR - Effect of High Energy on Further Recovery
 TEHEFR - Table for EHEFR

The individual's energy level, initialized at its maximum value of one, is affected by rates of depletion and recovery. The state of low energy known as exhaustion or fatigue is generally associated with other psychological problems, which may include irritability, sadness, detachment, disorientation, and low self-esteem.[35]

Energy is recovered during periods of leisure, relaxation, and, of course, sleep. The recovery rate is normalized at a point where the individual is working forty hours per week and has an energy level lower than .8; the recovery rate in this situation is 30% per week. But a longer workweek leaves less time for recovery and, therefore, a slower rate of recovery, as indicated in Figure 4.[36] This effect becomes proportionately




Figure 4. Table for Effect of Hours Worked on Energy Recovery

greater as the workweek increases to cut into evenings, weekends, and even late nights. When the energy level exceeds .8, the "effect of high energy on further recovery" acts to suppress the recovery rate somewhat, simply because not much lost energy remains to be recovered. This effect becomes stronger as the energy level approaches one, at which point the recovery rate must equal zero.

```

ED.KL = (EDN)(EPAED.K)(EHWED.K)(ELEFD.K)      16,A
EDN = .06                                       16.1,C

EPAED.K = TABLE(TEPAED,PAA.K,0,1.6,.2)       17,A
TEPAED = 5/4/3.1/2.3/1.6/1/.6/.4/.3          17.1,T

EHWED.K = TABLE(TEHWED,HWW.K,0,120,20)       18,A
TEHWED = .3/.6/1/1.5/2/2.5/3                 18.1,T

ELEFD.K = TABHL(TELEFD,EL.K,0,.2,.05)         19,A
TELEFD = 0/.4/.7/.9/1                         19.1,T

```

- ED - Energy Depletion (1/wk)
- EDN - Energy Depletion Normal
- EPAED - Effect of Perceived Adequacy on Energy Depletion
- TEPAED - Table for EPAED
- PAA - Perceived Adequacy of Accomplishment
- EHWED - Effect of Hours Worked on Energy Depletion
- TEHWED - Table for EHWED
- HWW - Hours Worked per Week (hours/wk)
- ELEFD - Effect of Low Energy on Further Depletion
- TELEFD - Table for ELEFD
- EL - Energy Level (0-1)

Energy is depleted as the result of repeated exposure to stress. The depletion rate is normalized at a point where the perceived adequacy of accomplishment is neutral, the workweek is forty hours, and the energy level is greater than .2; the depletion rate in this situation is only 6% per week. But should the work become frustrating or the hours at work increase, the normal stress of work can be greatly compounded, as illustrated in Figures 5 and 6.[37]

Frustration has been defined as the result of experiencing "undue delay in the fulfillment of a desired goal", [38] which may be interpreted here as a perceived inadequacy of accomplishment. Figure 5 indicates that as frustration increases, so too will its draining effect on energy. Figure 6 indicates that as the workweek increases, so will the exposure to work-related stressors, again resulting in faster depletion of energy. But note that energy depletion occurs even when the individual is not working at all, as a result of frustration, guilt, or boredom; [39] after all, "Ability brings with it the need to use




Figure 5. Table for Effect of Perceived Adequacy on Energy Depletion




Figure 6. Table for Effect of Hours Worked on Energy Depletion

463 that ability." [40] When the energy level falls below .2, the "effect of low energy on further depletion" acts to suppress the depletion rate somewhat, because not much energy remains to be lost. This effect becomes stronger as energy falls to zero, at which point the depletion rate must also equal zero.

MODEL BEHAVIOR

Description and Explanation of Baseline Behavior

The model's baseline behavior is presented in Figures 7 and 8, over a 75-week time horizon. During this time, the weekly accomplishment rate rises three times, only to fall precipitously along with the energy level as part of a self-sustaining burnout cycle. The observed limit cycle has a period of thirty weeks, one-third of this period being the decline phase during which accomplishment drops to one-sixth of its peak value.

The burnout cycle may be understood most clearly by referring to the causal-loop diagram presented in Figure 9. Initially, the perceived adequacy of accomplishment is relatively high, leading to increased expectations via Loop 2. Rising expectations keep the worker somewhat dissatisfied with his output, which drives the "work harder" response of Loop 1. Weekly accomplishment rises accordingly and continues to do so as Loops 1 and 2 combine to "bootstrap" the workweek upward toward its limit.

But as the hours worked increase, so does the stress of work, causing the energy level to decline. Falling energy puts a




Figure 7. Base Run: Accomplishment and Hours Worked per Week




Figure 8. Base Run: Energy Level and Perceived Adequacy of Accomplishment

464

dampener on output, leading to greater dissatisfaction and the beginnings of frustration. Dissatisfaction causes the individual to continue working hard, while frustration makes work even more stressful, both of which further speed the drain of energy. If the vicious cycles just described, seen in Figure 9 as positive loops 3 and 4, grow strong enough relative to the individual's ability to recover energy during non-work periods, then a collapse of energy and output like that seen in the base run will result.

Recovery from burnout is made possible by less work and reduced expectations. When energy falls to a low enough level, the individual finally breaks free of the addictive "work harder" response and reduces his hours at work; in the base run, hours worked per week fall rapidly from their peak of eighty to a trough of about forty.[41] This "time off" response finally stems the decline of energy, as suggested by Loop 5, and, indeed, allows lost energy to be reclaimed. While the workweek has been reduced, so have achievement expectations: Loop 2 now counters dissatisfaction by bringing output goals down to a level more in keeping with the exhausted worker's depressed capability.

The final stage of the cycle comes when modest expectations and rising energy enable the worker to achieve at a rate which is satisfying, without having to put in a lot of hours to do so. As seen in the base run, the perceived adequacy of accomplishment actually bounces back to exceed the neutral level of one, at a time when the individual is working a relatively short workweek.




Figure 9. Causal-loop Structure Underlying Burnout Cycles

But this euphoric period of "coasting" is short-lived, lasting only a couple of weeks (as corroborated by my own experience). Why? Because satisfaction is unnatural for the workaholic and only encourages him to expand his goals...and begin the cycle once again.[42]

Searching for Stability

In seeking to stabilize the limit cycle of burnout, it is instructive to examine first whether and under what conditions a stable equilibrium can be found. As it turns out, one can state the conditions for stable equilibrium with a single inequality comparing the forces affecting energy depletion with those affecting energy recovery. Interestingly, such an equilibrium

465

has the individual working steadily at his work limit. Taking the duration effects on depletion (TEHWED,18.1) and recovery (TEHWER,14.1) and the frustration effect on depletion (TEPAED,17.1) as givens, stability can be said to be threatened by: (1) a high depletion normal (EDN,16.1), (2) a low recovery normal (ERN,13.1), (3) a high upper limit on hours worked (LHWW,4.1), (4) a large bias to expand expectations (BFCX,10.1), and (5) downwardly inflexible expectations (TFCXPA,11.1). On the other hand, stability is independent of the steepness of both the "work harder" (TEPAHW,6.1) and "time off" (TEELHW,5.1) functions affecting hours worked per week.[43]

Model tests verify that the five factors mentioned above can be adjusted to stabilize the burnout cycle. The results of one such test, in which the energy depletion normal is reduced by 20%, from .06/week to .048/week, are shown in Figure 10. Although a limit cycle is still observed, its amplitude is considerably less than that of the base run (weekly accomplishment falls by less than 60%, compared with more than 80% in the base run), and its period is considerably longer (39 weeks versus 30). Another noticeable difference is that the decline phase now accounts for more than two-thirds of the entire cycle, compared with one-third in the base run. In effect, the vicious cycles causing burnout have been rendered less vicious, so that the decline is a slower and milder one, and one requiring less time off for recovery. As a result, weekly accomplishment averaged over the full 75 weeks of output is raised by 17% relative to the base run.[44]




Figure 10. Energy Depletion Normal Reduced by 20%

The results of this test are similar to the results of all those tests in which the stability condition described above is closer to being satisfied. The limit cycle becomes smaller in amplitude, longer in period, more drawn out in the decline phase, and faster in the recovery phase. The average workweek increases, as does average weekly accomplishment. If the stability condition is actually satisfied, the limit cycle is replaced by a critically-damped oscillation, with significantly higher average weekly accomplishment over the 75 weeks. For example, by reducing the energy depletion normal to .040 instead of .048 (a 33% reduction from the baseline instead of 20%), average weekly accomplishment is raised by 42% relative to the

466

base run (instead of 17%) and a consistent eighty-hour workweek is maintained. Thus, stabilizing policies do have the effect of increasing overall accomplishment, permitting the individual to work longer hours and to spend less time recuperating from exhaustion.

If the "time off" response to exhaustion does not affect the condition for stable equilibrium, one might wonder what effect this reasonable-sounding policy does have on behavior. Figure 11 shows the results of a run in which the individual is less reluctant than in the base run to break away from work during periods of reduced energy; in other words, the "time off" function falls more steeply than that seen in Figure 1.[45] As in the previous test, the limit cycle's amplitude is significantly reduced relative to the base run, a direct result of acting earlier to escape the vicious cycles of collapse. But this earlier response also leads to a cycle of shorter period, though similar in shape to the base run's cycle. The general impact is to shrink and speed up the cycle, rather than to stabilize it. In terms of overall output, though, this impact is beneficial; average weekly accomplishment is 12% higher in this test run than in the base run.

While a policy to reduce work hours in response to lower energy cannot stabilize the burnout cycle, a policy which reduces the limit on hours worked (LHWW) can; recall that this limit enters the condition for stable equilibrium. The essential difference between these two workweek policies is that the former




Figure 11. Steeper Effect of Energy Level on Hours Worked

is reactive, responding only when a problem has already surfaced, while the latter is proactive and preventive in nature. Also, as a practical matter, it may be easier to implement a shorter work limit (perhaps, as one author suggests, by locking the office doors after certain hours [46]) than to fight the workaholic's natural reluctance to take time off to relieve fatigue. Figure 12 shows the results of three test runs in which the work limit has been reduced from its baseline value of 80. When LHW=75, a limit cycle is still produced, though it has a smaller amplitude and longer period than the base run cycle. When LHW=72, the limit cycle is replaced by a damped oscillatory mode that achieves stable equilibrium by week 60, after an initial




Figure 12. Three Tests with Limit on Hours Worked per Week Reduced

overshoot and protracted undershoot. When LHW=69, the behavior is even more stable, consisting of a single small overshoot and virtual equilibrium soon after week 15. In terms of average accomplishment, stability again wins out; the first run improves on the base run by only 1%, the second run by 20%, and the third by 23%.

Figure 13 offers a broader view of the effect of the work limit (LHW) on overall results.[47] Three 75-week summary statistics--Average Accomplishments per Week to Date, Average Hours Worked per Week to Date, and Average Accomplishments per Hour to Date--are graphed against values of LHW ranging from 50




Figure 13. 75-week Summary Statistics as Affected by Work Limit

to 90 hours per week.[48] These graphs show clearly the difference between stability and instability, as far as total output and hours worked are concerned. Within the region of

468

stability (where $LHWW \leq 72.2$ hrs/wk [49]), a longer work limit means more hours worked (since $HWW=LHWW$ in equilibrium) but also lower hourly productivity (since longer hours reduce the energy level). Because the depressive effect of longer hours on average energy, small at first, accelerates as the region of instability is approached, a point of maximum average weekly accomplishment exists; given all other baseline parameter values, this "optimal work limit" is approximately 69 hours per week.

A tradeoff of hours and energy no longer exists once the region of instability is entered. A longer limit on hours worked does not increase the average number of hours worked in this region. Instead, its effect is to increase the amplitude and decrease the period of the limit cycle; that is, to destabilize the behavior further. In fact, changes in the work limit have little effect on any of the cumulative measures shown in Figure 13. It is only in the transition from stability to instability that major differences appear; both average accomplishment and average hours worked drop significantly when the burnout cycle is introduced. Burnout both reduces available energy and requires more time off from work. The advantages of stability are underscored by the following sort of comparison: In Figure 13, the individual can accomplish more on average with a work limit of 55 hrs/wk than with a work limit of 75 hrs/wk, while actually working fewer hours per week on average.

In noting that the optimal work limit is found within the region of stability but not far from the critical point of

transition to instability, one is led to an important conclusion: The more advantageous the other parameters affecting stability are, the higher the critical work limit will be, so the higher the optimal work limit will be as well. In concrete terms, this means that if the individual can (1) reduce stress at work (by finding ways to make the basic tasks more enjoyable, or by adopting a less pushy or more flexible approach to setting goals) or (2) relax more effectively when not working, then he will be able to work longer hours and accomplish more without risking burnout.

CONCLUSION

Probably the most effective first step in attempting to fight a complex problem like burnout is to understand its dynamic source. Burnout is not caused by a stressful work environment alone but, more importantly, by the individual's workaholic response to that environment. As many psychiatrists and counselors would agree, real progress begins only when the client has understood clearly his own role in creating his problems and his responsibility to adopt healthier behavior. I can personally attest to the powerful impact of (1) seeing my own burnout cycles reproduced by a computer model, (2) adjusting behavioral parameters until a stable solution was found, and (3) realizing then that my problem was not inevitable but, to a large degree, a product of my own work habits. It was particularly enlightening to discover that even from the point of view of productivity, psychological stability is preferable to instability.

The model suggests not only that stability is preferable to instability, but that it is generally attainable, even in a normally stressful work setting, through proper adjustment of one's work limit.[50] Chronic feelings of moderate tiredness or irritability at work must be met directly by a firm commitment to reduce one's maximum workweek permanently, or at least until the completion of a particularly stressful project. If the signs of incipient burnout later return even at this lower work limit, then the limit should be reduced again. It is important not to return to the old work limit, even when one is feeling rested and alert again; as the model demonstrates, vacations alone (the "time off" response) do not prevent a recurrence of burnout.

The model also demonstrates the potential benefit of reducing the normal stress at work or relaxing better when not working, through methods such as those described at the beginning of this paper. By adopting ways that make one less vulnerable to the negative consequences of stress, one expands the region of stability described in the previous section; this means that one can work longer hours without risking burnout and so accomplish more. An increased ability to (1) anticipate and counteract potential sources of stress, (2) relax without feeling guilty, or (3) adopt more realistic and flexible goals, can do much to make life as a high achiever more enjoyable and satisfying and return to the individual a sense of control.

NOTES

1. Oscar Wilde, quoted in Freudenberg p.52.
2. Greenwood p.xii.
3. Cherniss, Greenwood pp.117-163, Holt pp.427-433.
4. Greenwood pp.128-163, Ivancevich p.18.
5. Greenwood p.103, Holt pp.420-427, Ivancevich p.96.
6. Ivancevich p.171.
7. NDACTRD p.225.
8. NDACTRD p.1.
9. Freudenberg pp.13-16, Greenwood p.47, NDACTRD pp.1,125.
10. Greenwood p.126.
11. Greenwood pp.31-43, Ivancevich p.176, Selye pp.38-40.
12. Freudenberg pp.5-6.
13. Greenwood p.47, Selye pp.78,96.
14. Ivancevich p.96, NDACTRD p.151.
15. Ivancevich p.177, Machlowitz p.87, McClelland.
16. Freudenberg p.49, Machlowitz pp.27,122.
17. Machlowitz pp.103,117.
18. Freudenberg p.42, Machlowitz p.109, NDACTRD p.128.
19. Machlowitz pp.6,26-32,41-46.
20. Freudenberg pp.11,32, Machlowitz p.39, NDACTRD p.108.
21. Freudenberg p.186, see also p.198.
22. Freudenberg p.158, Machlowitz p.133, Selye p.40.
23. Cameron p.702, Freudenberg pp.158,175, Holroyd p.29, Ivancevich p.230, Janis p.82, NDACTRD p.192, Stoyva p.754.
24. Ivancevich pp.207-214, Kiefer p.6, Machlowitz pp.133-137, NDACTRD pp.192-202.
25. Freudenberg pp.136,206, Greenwood pp.175-211,229, Ivancevich pp.176,217, NDACTRD pp.186-191, Stoyva pp.749-753.

26. Freudenberg p.99, Greenwood p.171, Maslach.
27. Freudenberg p.210, Greenwood pp.220-221, Stoyva p.746.
28. Dabiri.
29. Greenwood p.125, Janis pp.69,99, Ivancevich p.201.
30. Ivancevich p.17.
31. Machlowitz pp.93-99.
32. Welford.
33. This response of accommodation is known in the stress literature as "syntoxic", in contrast with the "catatotoxic" or fighting response of working harder to meet one's goals. Both responses are considered homeostatic, because their purpose is to reduce potentially harmful stress. See Selye pp.41,47.
34. For values of PAA less than .8, $BFCX+FCXPA < 0$. Equilibrium can occur only when PAA=.8.
35. Freudenberg pp.17-18, Holt pp.427-433.
36. Breznitz p.5, Ivancevich p.176.
37. Greenwood pp.42,47, Howard, Ivancevich pp.77,176, Lazarus, NDACTRD p.1, Selye p.96.
38. Greenwood p.85.
39. Machlowitz p.114, Weiman, Selye p.73.
40. Albert Szent-Györgi, quoted in Selye p.73.
41. In this case of moderate burnout, an extended, full-time vacation is not needed.
42. In postscript to the above explanation of the burnout cycle, it should be noted that Loops 3 and 5--the "frustration" and "time off" loops--are not strictly necessary for generating the cycle. Model tests show that, if the work is normally disagreeable enough (i.e., if the energy depletion normal EDN is large enough), the draining power of long hours can generate a collapse without the added stress of mounting frustration. But in the baseline model (with its relatively small value of EDN), removal of the aggravating effect of Loop 3 does result in stable behavior. Other model tests show that when the individual does not respond to exhaustion by taking time off, his expectations eventually fall so low that he can satisfy them even while working less. The resulting respite from work permits a recovery of energy and

a renewal of higher accomplishment. But the removal of Loop 5 does delay the recovery, resulting in a considerably more severe and protracted period of burnout.

- 43. Equilibrium requires that hours worked per week (HWW), expectations of weekly accomplishment (XAW), and the energy level (EL) all be unchanging. An equilibrium at which HWW is less than its limit (LHWW) can be shown to exist under certain circumstances; but this equilibrium is unstable, in the sense that any small perturbation of certain exogenous parameters will trigger a limit cycle. The condition for a stable equilibrium requires that $HWW=LHWW$. Labeling EHWED as f, EHWER as g, and EPAED as h, the condition may be stated as follows:

$$EDN * h(PAA') * f(LHWW) \leq ERN * g(LHWW)$$

$$\text{where } FCXPA(PAA') = -BFCX$$

In the baseline model, the left-hand side of the inequality equals .19, while the right-hand side equals .15: The condition for stable equilibrium is not met and a limit cycle results.

- 44. Average Accomplishments per Week to Date, or AVAWD, increases from the baseline value of 47.8 to 55.1.
- 45. In the test run, $TEELHW=0/.3/.5/.7/.9/1$, compared to the base run's $TEELHW=0/.4/.7/.9/1/1$.
- 46. Machlowitz p.133.
- 47. This figure summarizes the results of a number of model tests, like those in Figure 12, in which the work limit was altered. Recall that $LHWW=80$ in the base run.
- 48. DYNAMO equations for these three summary statistics may be found in the Appendix, listed as AVAWD, AVHWWD, and AVAHD, respectively. AVAWD is computed by integrating Accomplishments per Week (AW) over time and then dividing by the total time elapsed. AVHWWD is similarly computed by integrating Hours Worked per Week (HWW) and then dividing by time elapsed. AVAHD is found by dividing cumulative accomplishments (Accomplishments to Date, AD) by cumulative hours worked (Hours Worked to Date, HWD).
- 49. The analytic stability condition presented above is satisfied in this region. When $LHWW=72.2$, the left-hand (depletion) side equals the right-hand (recovery) side.
- 50. The person-environment fit may conceivably be so poor that the optimal work limit is less than a standard workweek or even nonexistent. In such cases, an individual in search of full-time work is best advised to find it elsewhere. Why court disaster?

REFERENCES

Breznitz S, Goldberger L. "Stress research at a crossroads." In Handbook of Stress, ed. L Goldberger, S Breznitz. The Free Press: NY, 1982.

Cameron R, Meichenbaum D. "The nature of effective coping and the treatment of stress related problems: A cognitive-behavioral perspective." In Handbook of Stress, 1982.

Cherniss C. Staff Burnout: Job Stress in the Human Services. Sage: Beverly Hills, 1980.

Dabiri HE. "Dynamics of human development: Achievement crisis." AP Sloan School Working Paper WP 1043-79, Massachusetts Institute of Technology, 1979.

Freudenberger HJ, Richelson G. Burn-Out: The High Cost of High Achievement. Anchor Press: Garden City NY, 1980.

Greenwood JW III, Greenwood JW Jr. Managing Executive Stress. Wiley and Sons: NY, 1979.

Holroyd KA, Lazarus RS. "Stress, coping, and somatic adaptation." In Handbook of Stress, 1982.

Holt RR. "Occupational Stress." In Handbook of Stress, 1982.

Howard A, Scott RA. "A proposed framework for the analysis of stress in the human organism." Behavioral Science, 10(2), 1965.

Ivancevich JM, Matteson MT. Stress and Work. Scott, Foresman: Illinois, 1980.

Janis IL. "Decision-making under stress." In Handbook of Stress, 1982.

Kiefer C, Senge PM. "Metanoic organizations in the transition to a sustainable society." MIT System Dynamics Group Memo D-3360-3 (presented at the 1982 Woodlands Conference), 1982.

Lazarus RS. "Positive denial: The case for not facing reality." Psychology Today, 44-60, November 1979.

Machlowitz MM. Workaholics: Living with Them, Working with Them. Addison-Wesley: Massachusetts, 1980.

Maslach C. "Job burnout: How people cope." Public Welfare, 36(2), 1978.

McClelland DC. The Achieving Society. Van Nostrand Reinhold: Princeton, 1961.

NDACTRD (National Drug Abuse Center for Training and Resource Development). Trainer Manual: Staff Burnout. NDACTRD Publication No. 80-00115, Washington DC, 1980.

Selye H. Stress without Distress. Lippincott: Philadelphia, 1974.

Stoyva J, Anderson C. "A coping-rest model of relaxation and stress management." In Handbook of Stress, 1982.

Weiman CG. "A study of occupational stressors and the incidence of disease/risk." Journal of Occupational Medicine, 19(2): 119-122, 1977.

Welford AT. "Stress and performance." Ergonomics, 16:567-580, 1973.

APPENDIX: WORKER BURNOUT MODEL LISTING

* WORKER BURNOUT CYCLES

NOTE BY J.B. HOMER, APRIL 1984

- 1 A AW.K=(AH.K)(HWW.K)
- 2 A AH.K=TABLE(TAH,EL.K,0,1,.2)
- T TAH=0/.2/.4/.6/.8/1 A-UNITS/HR
- 3 L HWW.K=HWW.J+(DT/TAHWW)(IHWW.J-HWW.J)
- N HWW=HWWI
- C HWWI=40 HRS/WK
- C TAHWW=1 WK
- 4 A IHWW.K=MIN(LHWW,HWW.K*EELHW,K*EPAHW.K)
- C LHWW=80 HRS/WK
- 5 A EELHW.K=TABLE(TEELHW,EL.K,0,1,.2)
- T TEELHW=0/.4/.7/.9/1/1
- 6 A EPAHW.K=TABLE(TEPAHW,PAA.K,0,1.6,.2)
- T TEPAHW=2.3/1.9/1.6/1.35/1.15/1/.9/.8/.75
- 7 A PAA.K=PAW.K/XAW.K
- 8 L PAW.K=PAW.J+(DT/TPAW)(AW.J-PAW.J)
- N PAW=AW
- C TPAW=1 WK
- 9 L XAW.K=XAW.J+(DT)(XAW.J*FCXAW.J)
- N XAW=XAWI
- C XAWI=40 A-UNITS/WK
- 10 A FCXAW.K=BFCX+FCXPA.K
- C BFCX=.1 PER WK
- 11 A FCXPA.K=TABLE(TFCXPA,PAA.K,0,1.6,.2)
- T TFCXPA=-.7/-.5/-.35/-.2/-.1/0/.1/.25/.4
- 12 L EL.K=EL.J+(DT)(ER.JK-ED.JK)
- N EL=ELI
- C ELI=1
- 13 R ER.KL=(ERN)(EHWER.K)(EHEFR.K)
- C ERN=.3 PER WK
- 14 A EHWER.K=TABLE(TEHWER,HWW.K,0,120,20)
- T TEHWER=1.3/1.2/1/.7/.5/.35/.25
- 15 A EHEFR.K=TABHL(TEHEFR,EL.K,.8,1,.05)
- T TEHEFR=1/.9/.7/.4/0
- 16 R ED.KL=(EDN)(EPAED.K)(EHWED.K)(ELEFD.K)
- C EDN=.06 PER WK
- 17 A EPAED.K=TABLE(TEPAED,PAA.K,0,1.6,.2)
- T TEPAED=5/4/3.1/2.3/1.6/1/.6/.4/.3
- 18 A EHWED.K=TABLE(TEHWED,HWW.K,0,120,20)
- T TEHWED=.3/.6/1/1.5/2/2.5/3
- 19 A ELEFD.K=TABHL(TELEFD,EL.K,0,.2,.05)
- T TELEFD=0/.4/.7/.9/1
- NOTE SUMMARY STATISTICS
- 20 A AVAWD.K=AD.K/(TIME.K+1E-7)
- 21 L AD.K=AD.J+(DT)(AW.J)
- N AD=AW*1E-7
- 22 A AVHWD.K=HWD.K/(TIME.K+1E-7)
- 23 L HWD.K=HWD.J+(DT)(HWW.J)
- N HWD=HWW*1E-7
- 24 A AVAHD.K=AD.K/HWD.K

473

```
NOTE CONTROL STATEMENTS
SPEC DT=.25/LENGTH=75/PLTPER=1.5/PRTPER=15
PRINT AVAWD,AVHWWD,AVAHD
PLOT AW=A,HWW=H(0,100)
PLOT EL=E,PAA=P(0,1.6)
RUN BASE
```